
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 22.5.14 which is part of the NBIOT test suite. Only essential changes to the TTCN are applied and documented in section 4.

With these changes applied the test case can be demonstrated to run with one or more NBIOT UEs 
(see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Case:
TC_22_5_14
Test Group:
/NBIOT/22_2/NBIOT_Idle_CellReselection.ttcn

ATS Version:
iwd-TTCN3-B2013-03_D17wk51 + essential modifications.

System Simulator used:
Rohde & Schwarz CMW500 / Anritsu PCT ME7834LA
UE used:
Hisilicon Hi2110/ MTK_2625
Verification Status:
PASS
4 Corrections required for test case 22.5.14
4.1 Introduction

This section describes the changes required to make test case 22.5.14 run correctly with a NBIOT UE. All modifications are marked with label “WA#” in the TTCN comments column of the enclosed ATS [1].

The ATS version used as basis was NBIOT testsuite which is part of iwd-TTCN3-B2013-03_D17wk51 release. This testsuite provided by MCC160 contains NBIOT test cases. 
4.2 Change  – function fl_TC_22_5_14_Body()
	Function name
	fl_TC_22_5_14_Body()

	Reason for change
	In the current TTCN implementation :
1) For step 29B-29G the function f_NBIOT_UE_Detach_SwitchOffUe(nbiot_Cell50, CONTROL_PLANE, RRC_IDLE, cr_DetachTypeOnSwitchOff); is called. This uses the template  cr_DetachTypeOnSwitchOff, which allows any type of detacy type. So, a device which sends a detacy type set as IMSI detach, or Combined EPS/IMSI detach will also pass. This should be avoided as a NBIOT device should only be transmitting a EPS only detach.
2) At step 32, the TTCN expects the UE to transmit the GUTI assigned in step 19-29. However, as per 24.301 Sec 5.5.1.2.2, if the UE is trying to register to a PLMN which is not in EPLMN list, or is not registered PLMN, then the UE will use IMSI. The TTCN needs to be corrected.
3) At step 38, according to the current Prose , the power setting T6 is applied. This sets the Ncell 55 to serving cell (-85dBm), and Ncell 56 to -97dBm. At step 39 it is expected that the UE will perform an ATTACH on Ncell 56. Considering that both, Ncell 55 and Ncell 56 are on same frequency, it is very difficult for the UE to reselect to Ncell 56 particularly when it is at a low power as well. It is therefore proposed that Ncell 56 be set to -91dBm, and Ncell 55 to -97dBm at this stage, in order to make cell reselection to Ncell 56 easy for the device. This is also not going to impact any test purpose requirement either, but will reduce significantly the time taken by the UE to reselect on Ncell 56. A Prose will be raised for this.
4) At step 64, the template used for ATTACH REQ expects the nasKeysetIdentifier to be set to ‘111’B, which means no key. This is incorrect, and needs to be changed. Also, this message is expected as unprotected. This also should be changed to integrityprotected.
5) At Step 69, the function f_NBIOT_CheckNoAttachOn2Cells(nbiot_Cell55, nbiot_Cell62, 30.0, "Step 67"); is called. This should be assigning a verdict for Step 69, not Step 67.
6) At Step 74, power setting T13 is applied, according to which Ncell 55 is changed to -91dBm, and Ncell 56 to -85dBm. This is after the UE has transmitted the ATTACH REQ on Ncell 55, but has NOT received the ATTACH REJECT from the SS on Ncell 56. So, the UE is expected to transmit the ATTACH on Ncell55, then the SS applies T13 and performs Integrity protection on Ncell 55 after the reception of ATTACH REQ, and then transmit the ATTACH REJ on Ncell 56. The expectation is that the UE will have performed a cell reselection before the SS transmits the ATTACH REJ. This in theory is possible, but not all UEs can be that quick. This will make the testcase unstable. It is therefore proposed to move the applying T13 after the SS has transmitted the ATTACH REJ. This will have no impact on test purpose, and make the testcase stable. A Prose CR will be raised for this.
7) In the current TTCN, the Steps 74A-74D and Step 75 are being performed on Ncell 56. These should be performed on Ncell 55 instead.
8) At step 80, the current TTCN still tries to set Ncell 50 to suitable neighbour power. However, this cell has been release earlier.

	Summary of change
	1) Instead of using f_NBIOT_UE_Detach_SwitchOffUe(nbiot_Cell50, CONTROL_PLANE, RRC_IDLE, cr_DetachTypeOnSwitchOff);, used f_NBIOT_UE_Detach_SwitchOffUe(nbiot_Cell50, CONTROL_PLANE, RRC_IDLE);
2) For Step 32, in the ATTACH REQ, instead of using f_GutiParameters2MobileIdentity(omit, v_Guti1_Params), used f_Imsi2MobileIdentity(px_IMSI_Def) instead.
3) Set the Ncell 56 to -91dBm, and Ncell 55 to -97dBm.
4) Instead of using the template cdr_CIOT_ATTACH_REQUEST_WithIMSINoTai, used cr_CIOT_ATTACH_REQUEST. And used tsc_SHT_IntegrityProtected instead of tsc_SHT_NoSecurityProtection.
5) Updated the function call to f_NBIOT_CheckNoAttachOn2Cells(nbiot_Cell55, nbiot_Cell62, 30.0, "Step 69");
6) Moved the applying of T13 after Step 76. Please see below.
7) Instead of using nbiot_Cell56, used nbiot_Cell55 instead fot Steps 74A-74D and Step 75.
8) At step 80, removed the code which sets the Ncell 50 to suitable neighbour cell.

	TTCN module
	NBIOT_NAS_Attach.ttcn

	MCC160 Comment
	


Before change
	function fl_TC_22_5_14_Body() runs on NBIOT_PTC
  {
    var NAS_MSG_Indication_Type v_NasInd;
    var NasCount_Type v_NasCountUL;
    var GutiParameters_Type v_Guti1_Params := f_NBIOT_CellInfo_GetGuti(nbiot_Cell50);
    var NAS_PlmnId v_PLMN1 := f_Asn2Nas_PlmnId(v_Guti1_Params.PLMN_Identity);
    var template (present) B3_Type v_PDNType := (f_GetPdnType(), '101'B);
    var template (value) NB_CellPowerList_Type v_CellPowerList;
    var template (present) NAS_UL_Message_Type v_ESMMessage;
    if (px_DoAttachWithoutPDN) {
      v_ESMMessage := cr_ESM_DUMMY_MESSAGE;
    } else {
      v_ESMMessage := cr_CIOT_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach, v_PDNType);
    }
    //@siclog "Step 1" siclog@
    //SS configures adjusts the cell power levels according to row T1 in table 22.5.14.3-1:
….

….

    //@siclog "Step 19 - 29" siclog@
    //The attach procedure is completed by executing steps 5 to 14b1 of the UE registration procedure in TS 36.508 sub clause 8.1.5.2.
    f_NBIOT_InitialRegistration_Step5_14(nbiot_Cell50, CONTROL_PLANE, STATE2_NB_CONNECTED_UPDATED, v_NasInd);
    //@siclog "Step 29A" siclog@
    //The SS transmits an RRCConnectionRelease-NB message to release RRC connection and move to RRC_IDLE.
    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell50);
    //@siclog "Step 29B - 29G" siclog@
    //The UE is switched off or the USIM is removed.
    //The UE transmits an RRCConnectionRequest-NB message.
    //SS transmits an RRCConnectionSetup-NB message.
    //The UE transmits an RRCConnectionSetupComplete-NB message.
    //The UE transmits a DETACH REQUEST message with the Detach Type IE indicating "switch off".
    f_NBIOT_UE_Detach_SwitchOffUe(nbiot_Cell50, CONTROL_PLANE, RRC_IDLE, cr_DetachTypeOnSwitchOff);
    //@siclog "Step 30" siclog@
    //SS adjusts the cell power levels according to row T5 in table 22.5.14.3-1
    v_CellPowerList := {
      cs_NB_CellPower(nbiot_Cell50, tsc_NonSuitableOffNBIOTCellRS_EPRE),
      cs_NB_CellPower(nbiot_Cell51, tsc_NonSuitableOffNBIOTCellRS_EPRE),
      cs_NB_CellPower(nbiot_Cell55, tsc_ServingCellRS_EPRE),
      cs_NB_CellPower(nbiot_Cell56, tsc_NonSuitableOffNBIOTCellRS_EPRE),
      cs_NB_CellPower(nbiot_Cell62, tsc_NonSuitableOffNBIOTCellRS_EPRE)
    };
    f_NBIOT_SetCellPowerList(v_CellPowerList);
    f_NBIOT_ReleaseCellSpecific(nbiot_Cell51);//NBIOT Cell to be released as soon as it is not used anymore
    //The following messages are to be observed on Ncell 55 unless explicitly stated otherwise.
    //@siclog "Step 31" siclog@
    //The UE is switched on.
    f_UT_SwitchOnUE(UT, false);
    //@siclog "Step 32" siclog@
    //The UE transmits an ATTACH REQUEST message. IF AttachWithoutPDN AND UE sets attachWithoutPDN-Connectivity? in RRCConnectionSetupComplete-NB
    //THEN the ESM DUMMY MESSAGE is piggybacked in ATTACH REQUEST, OTHERWISE the PDN CONNECTIVITY REQUEST message is piggybacked in ATTACH REQUEST.
    v_NasInd := f_NBIOT_RRC_ConnEst_DefWithNas(nbiot_Cell55,
                                               cr_EstablishmentCause_NB_Any,
                                               cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected,
                                                                                 cdr_CIOT_ATTACH_REQUEST_WithGuti(cr_AdditionalUpdateType_CIOT,
                                                                                                                  f_NBIOT_SecurityKSIasme_Get(),
                                                                                                                  f_GutiParameters2MobileIdentity(omit, v_Guti1_Params), // GUTI1 @sic R5s170254 sic@
                                                                                                                  cr_TAI(v_PLMN1, tsc_TAC1)),
                                                                                 v_ESMMessage));
    //@siclog "Step 33" siclog@
…
…
    //@siclog "Step 37" siclog@
    //Check: Does the UE transmit an ATTACH REQUEST message in the next 30 seconds?
    f_NBIOT_CheckNoAttachOn1Cell(nbiot_Cell55, 30.0, "Step 37");
    //@siclog "Step 38" siclog@
    //SS adjusts the cell power levels according to row T6 in table 22.5.14.3-1
    //NCell51 will not be config power level since it is not used anymore
    v_CellPowerList := {
      cs_NB_CellPower(nbiot_Cell50, tsc_NonSuitableOffNBIOTCellRS_EPRE),
      cs_NB_CellPower(nbiot_Cell55, tsc_ServingCellRS_EPRE),
      cs_NB_CellPower(nbiot_Cell56, tsc_SuitableCellRS_EPRE),
      cs_NB_CellPower(nbiot_Cell62, tsc_NonSuitableOffNBIOTCellRS_EPRE)
    };
    f_NBIOT_SetCellPowerList(v_CellPowerList);
    //@siclog "Step 39" siclog@
…
…

    //@siclog "Step 63" siclog@
    //The UE is brought back to operation or the USIM is inserted.
    f_UT_SwitchOnUE(UT, true);
    //The following messages are to be observed on Ncell 55 unless explicitly stated otherwise.
    //@siclog "Step 64" siclog@
    //The UE transmits an ATTACH REQUEST message?
    //IF AttachWithoutPDN AND UE sets attachWithoutPDN-Connectivity? in RRCConnectionSetupComplete-NB
    //THEN the ESM DUMMY MESSAGE is piggybacked in ATTACH REQUEST, OTHERWISE the PDN CONNECTIVITY REQUEST message is piggybacked in ATTACH REQUEST.
    v_NasInd := f_NBIOT_RRC_ConnEst_DefWithNas(nbiot_Cell55,
                                               cr_EstablishmentCause_NB_Any,
                                               cr_NAS_IndicationWithPiggybacking(tsc_SHT_NoSecurityProtection,
                                                                                 cdr_CIOT_ATTACH_REQUEST_WithIMSINoTai(cr_AdditionalUpdateType_CIOT),
                                                                                 v_ESMMessage));
    //@siclog "Step 65" siclog@
    //The SS transmits an ATTACH REJECT message, EMM cause = " No suitable cells in tracking area ".
    //(The list of "forbidden tracking areas for roaming" in the UE should now contain TAI-7)
    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell55,
                                   tsc_SRB1bis,
                                   cs_TimingInfo_Now,
                                   cs_NAS_Request(f_GetEPS_SecurityHeaderForReject(tsc_Cause_NoSuitableCells), // @sic R5-174501 sic@
                                                  cs_508_AttachReject(cs_EMM_Cause_v(tsc_Cause_NoSuitableCells)))));
    //@siclog "Step 65A" siclog@
…

…

    //@siclog "Step 69" siclog@
    //Check: Does the UE transmit an ATTACH REQUEST message in the next 30 seconds on Ncell 55 or Ncell 62?
    f_NBIOT_CheckNoAttachOn2Cells(nbiot_Cell55, nbiot_Cell62, 30.0, "Step 67");
    //@siclog "Step 70" siclog@
    //If possible (see ICS) switch off is performed or the USIM is removed.
    //Otherwise the power is removed.
    f_UT_SwitchOffUE(UT, true);
    //@siclog "Step 71" siclog@
    //SS adjusts the cell power levels according to row T12 in table 22.5.14.3-1
    //NCell50 and NCell51 will not be config power level since it is not used anymore
    v_CellPowerList := {
      cs_NB_CellPower(nbiot_Cell55, tsc_ServingCellRS_EPRE),
      cs_NB_CellPower(nbiot_Cell56, tsc_NonSuitableOffNBIOTCellRS_EPRE),
      cs_NB_CellPower(nbiot_Cell62, tsc_NonSuitableOffNBIOTCellRS_EPRE)
    };
    f_NBIOT_SetCellPowerList(v_CellPowerList);
    f_NBIOT_ReleaseCellSpecific(nbiot_Cell62);//NBIOT Cell to be released as soon as it is not used anymore
    //@siclog "Step 72" siclog@
    //The UE is brought back to operation or the USIM is inserted.
    f_UT_SwitchOnUE(UT, true);
    //@siclog "Step 73" siclog@
    //The UE transmits an ATTACH REQUEST message on Ncell 55.
    //IF AttachWithoutPDN AND UE sets attachWithoutPDN-Connectivity? in RRCConnectionSetupComplete-NB
    //THEN the ESM DUMMY MESSAGE is piggybacked in ATTACH REQUEST, OTHERWISE the PDN CONNECTIVITY REQUEST message is piggybacked in ATTACH REQUEST.
    v_NasInd := f_NBIOT_RRC_ConnEst_DefWithNas(nbiot_Cell55,
                                               cr_EstablishmentCause_NB_Any,
                                               cr_NAS_IndicationWithPiggybacking(tsc_SHT_NoSecurityProtection,
                                                                                 cdr_CIOT_ATTACH_REQUEST_WithIMSINoTai(cr_AdditionalUpdateType_CIOT),
                                                                                 v_ESMMessage));
    //@siclog "Step 74" siclog@
    //SS adjusts the cell power levels according to row T13 in table 22.5.14.3-1
    //NCell50 and NCell51 and NCell62 will not be config power level since it is not used anymore
    v_CellPowerList := {
      cs_NB_CellPower(nbiot_Cell55, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE),
      cs_NB_CellPower(nbiot_Cell56, tsc_ServingCellRS_EPRE)
    };
    f_NBIOT_SetCellPowerList(v_CellPowerList);
    //@siclog "Step 74A - 74D" siclog@
    //Steps 5 to 8 from procedure 8.1.5.2.3 in TS 36.508.
    v_NasCountUL := f_NBIOT_InitialRegistration_Step5_8 (nbiot_Cell56,
                                                         v_NasInd,
                                                         false);//@sic R5-176875 sic@
    //@siclog "Step 75" siclog@
    //The SS transmits an ATTACH REJECT message,
    //EMM cause = No suitable cells in tracking area?.
    //(The list of forbidden tracking areas for roaming? in the UE should now contain TAI-9)
    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell55,
                                   tsc_SRB1bis,
                                   cs_TimingInfo_Now,
                                   cs_NAS_Request(f_GetEPS_SecurityHeaderForReject(tsc_Cause_NoSuitableCells), // @sic R5-174501 sic@
                                                  cs_508_AttachReject(cs_EMM_Cause_v(tsc_Cause_NoSuitableCells)))));
    //@siclog "Step 76" siclog@
    //The SS releases the RRC connection.
    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell55);
    //@siclog "Step 77" siclog@
    //Check: Does the UE transmit an ATTACH REQUEST message on Ncell 56?
    //IF AttachWithoutPDN AND UE sets attachWithoutPDN-Connectivity? in RRCConnectionSetupComplete-NB
    //THEN the ESM DUMMY MESSAGE is piggybacked in ATTACH REQUEST, OTHERWISE the PDN CONNECTIVITY REQUEST message is piggybacked in ATTACH REQUEST.
    v_NasInd := f_NBIOT_RRC_ConnEst_DefWithNas(nbiot_Cell56,
                                               cr_EstablishmentCause_NB_Any,
                                               cr_NAS_IndicationWithPiggybacking(tsc_SHT_NoSecurityProtection,
                                                                                 cdr_CIOT_ATTACH_REQUEST_WithIMSINoTai(cr_AdditionalUpdateType_CIOT),
                                                                                 v_ESMMessage));
    //@siclog "Step 77A - 77D" siclog@
    //Steps 5 to 8 from procedure 8.1.5.2.3 in TS 36.508.
    v_NasCountUL := f_NBIOT_InitialRegistration_Step5_8 (nbiot_Cell56,
                                                         v_NasInd,
                                                         false);//@sic R5-176875 sic@
    //@siclog "Step 78" siclog@
    //The SS transmits an ATTACH REJECT message, EMM cause = No suitable cells in tracking area?.
    //(The list of forbidden tracking areas for roaming? in the UE should now contain TAI-7 and TAI-8)
    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell56,
                                   tsc_SRB1bis,
                                   cs_TimingInfo_Now,
                                   cs_NAS_Request(f_GetEPS_SecurityHeaderForReject(tsc_Cause_NoSuitableCells), // @sic R5-174501 sic@
                                                  cs_508_AttachReject(cs_EMM_Cause_v(tsc_Cause_NoSuitableCells)))));
    //@siclog "Step 79" siclog@
    //The SS releases the RRC connection.
    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell56);
    //@siclog "Step 80" siclog@
    //SS adjusts the cell power levels according to row T14 in table 22.5.14.3-1
    //NCell50 and NCell51 and NCell62 will not be config power level since it is not used anymore
    v_CellPowerList := {
      cs_NB_CellPower(nbiot_Cell50, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE),//@sic R5-176875 sic@
      cs_NB_CellPower(nbiot_Cell55, tsc_ServingCellRS_EPRE),
      cs_NB_CellPower(nbiot_Cell56, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE)
    };
    f_NBIOT_SetCellPowerList(v_CellPowerList);
    //The following messages are to be observed on Ncell 55 unless explicitly stated otherwise.
    //@siclog "Step 81" siclog@
    //Check: Does the UE transmit the ATTACH REQUEST message in the next 30 seconds?
    f_NBIOT_CheckNoAttachOn1Cell(nbiot_Cell55, 30.0, "Step 81");
    //@siclog "Step 82" siclog@
    //If possible (see ICS) switch off is performed. Otherwise the power is removed.
    f_UT_SwitchOffUE(UT, true);
    //@siclog "Step 83" siclog@
    //The UE is brought back to operation.
    f_UT_SwitchOnUE(UT, true);
    //@siclog "Step 84" siclog@
    //Check: Does the UE transmit an ATTACH REQUEST?
    //IF AttachWithoutPDN AND UE sets attachWithoutPDN-Connectivity? in RRCConnectionSetupComplete-NB
    //THEN the ESM DUMMY MESSAGE is piggybacked in ATTACH REQUEST, OTHERWISE the PDN CONNECTIVITY REQUEST message is piggybacked in ATTACH REQUEST.
    v_NasInd := f_NBIOT_RRC_ConnEst_DefWithNas(nbiot_Cell55,
                                               cr_EstablishmentCause_NB_Any,
                                               cr_NAS_IndicationWithPiggybacking(tsc_SHT_NoSecurityProtection,
                                                                                 cdr_CIOT_ATTACH_REQUEST_WithIMSINoTai(cr_AdditionalUpdateType_CIOT),
                                                                                 v_ESMMessage));
    //@siclog "Step 85 - 95" siclog@
    //The attach procedure is completed by executing steps 5 to 14b1 of the UE registration procedure in TS 36.508 sub clause 8.1.5.2.
    f_NBIOT_InitialRegistration_Step5_14(nbiot_Cell55, CONTROL_PLANE, STATE2_NB_CONNECTED_UPDATED, v_NasInd);
    //At the end of this test procedure sequence, the UE is in end state 2-NB Connected Mode according to TS 36.508.
  }


After change
	function fl_TC_22_5_14_Body() runs on NBIOT_PTC
  {
    var NAS_MSG_Indication_Type v_NasInd;
    var NasCount_Type v_NasCountUL;
    var GutiParameters_Type v_Guti1_Params := f_NBIOT_CellInfo_GetGuti(nbiot_Cell50);
    var NAS_PlmnId v_PLMN1 := f_Asn2Nas_PlmnId(v_Guti1_Params.PLMN_Identity);
    var template (present) B3_Type v_PDNType := (f_GetPdnType(), '101'B);
    var template (value) NB_CellPowerList_Type v_CellPowerList;
    var template (present) NAS_UL_Message_Type v_ESMMessage;
    if (px_DoAttachWithoutPDN) {
      v_ESMMessage := cr_ESM_DUMMY_MESSAGE;
    } else {
      v_ESMMessage := cr_CIOT_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach, v_PDNType);
    }
    //@siclog "Step 1" siclog@
    //SS configures adjusts the cell power levels according to row T1 in table 22.5.14.3-1:
...

...

    //@siclog "Step 19 - 29" siclog@
    //The attach procedure is completed by executing steps 5 to 14b1 of the UE registration procedure in TS 36.508 sub clause 8.1.5.2.
    f_NBIOT_InitialRegistration_Step5_14(nbiot_Cell50, CONTROL_PLANE, STATE2_NB_CONNECTED_UPDATED, v_NasInd);
    //@siclog "Step 29A" siclog@
    //The SS transmits an RRCConnectionRelease-NB message to release RRC connection and move to RRC_IDLE.
    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell50);    
    //@siclog "Step 29B - 29G" siclog@
    //The UE is switched off or the USIM is removed.
    //The UE transmits an RRCConnectionRequest-NB message.
    //SS transmits an RRCConnectionSetup-NB message.
    //The UE transmits an RRCConnectionSetupComplete-NB message.
    //The UE transmits a DETACH REQUEST message with the Detach Type IE indicating "switch off".
    //WA#22_5_14 f_NBIOT_UE_Detach_SwitchOffUe(nbiot_Cell50, CONTROL_PLANE, RRC_IDLE, cr_DetachTypeOnSwitchOff);
    f_NBIOT_UE_Detach_SwitchOffUe(nbiot_Cell50, CONTROL_PLANE, RRC_IDLE); //WA#22_5_14    
    //@siclog "Step 30" siclog@
    //SS adjusts the cell power levels according to row T5 in table 22.5.14.3-1
    v_CellPowerList := {
      cs_NB_CellPower(nbiot_Cell50, tsc_NonSuitableOffNBIOTCellRS_EPRE),
      cs_NB_CellPower(nbiot_Cell51, tsc_NonSuitableOffNBIOTCellRS_EPRE),
      cs_NB_CellPower(nbiot_Cell55, tsc_ServingCellRS_EPRE),
      cs_NB_CellPower(nbiot_Cell56, tsc_NonSuitableOffNBIOTCellRS_EPRE),
      cs_NB_CellPower(nbiot_Cell62, tsc_NonSuitableOffNBIOTCellRS_EPRE)
    };
    f_NBIOT_SetCellPowerList(v_CellPowerList);
    f_NBIOT_ReleaseCellSpecific(nbiot_Cell51);//NBIOT Cell to be released as soon as it is not used anymore
    //The following messages are to be observed on Ncell 55 unless explicitly stated otherwise.
    //@siclog "Step 31" siclog@
    //The UE is switched on.    
    f_UT_SwitchOnUE(UT, false);
    //@siclog "Step 32" siclog@
    //The UE transmits an ATTACH REQUEST message. IF AttachWithoutPDN AND UE sets attachWithoutPDN-Connectivity? in RRCConnectionSetupComplete-NB
    //THEN the ESM DUMMY MESSAGE is piggybacked in ATTACH REQUEST, OTHERWISE the PDN CONNECTIVITY REQUEST message is piggybacked in ATTACH REQUEST.
    v_NasInd := f_NBIOT_RRC_ConnEst_DefWithNas(nbiot_Cell55,
                                               cr_EstablishmentCause_NB_Any,
                                               cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected,
                                                                                 cdr_CIOT_ATTACH_REQUEST_WithGuti(cr_AdditionalUpdateType_CIOT,
                                                                                                                  f_NBIOT_SecurityKSIasme_Get(),
                                                                                                                  //f_GutiParameters2MobileIdentity(omit, v_Guti1_Params), // GUTI1 @sic R5s170254 sic@
                                                                                                                  f_Imsi2MobileIdentity(px_IMSI_Def),//WA#22_5_14
                                                                                                                  cr_TAI(v_PLMN1, tsc_TAC1)),
                                                                                 v_ESMMessage));
    //@siclog "Step 33" siclog@
…
…

    //@siclog "Step 37" siclog@
    //Check: Does the UE transmit an ATTACH REQUEST message in the next 30 seconds?
    f_NBIOT_CheckNoAttachOn1Cell(nbiot_Cell55, 30.0, "Step 37");
    //@siclog "Step 38" siclog@
    //SS adjusts the cell power levels according to row T6 in table 22.5.14.3-1
    //NCell51 will not be config power level since it is not used anymore
    v_CellPowerList := {
      cs_NB_CellPower(nbiot_Cell50, tsc_NonSuitableOffNBIOTCellRS_EPRE),
      cs_NB_CellPower(nbiot_Cell55, tsc_SuitableCellRS_EPRE /*WA#22_5_14 tsc_ServingCellRS_EPRE*/),
      cs_NB_CellPower(nbiot_Cell56, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE /*WA#22_5_14 tsc_SuitableCellRS_EPRE*/),
      cs_NB_CellPower(nbiot_Cell62, tsc_NonSuitableOffNBIOTCellRS_EPRE)
    };
    f_NBIOT_SetCellPowerList(v_CellPowerList);
    //@siclog "Step 39" siclog@
…

…

    //@siclog "Step 63" siclog@
    //The UE is brought back to operation or the USIM is inserted.
    f_UT_SwitchOnUE(UT, true);
    //The following messages are to be observed on Ncell 55 unless explicitly stated otherwise.
    //@siclog "Step 64" siclog@
    //The UE transmits an ATTACH REQUEST message?
    //IF AttachWithoutPDN AND UE sets attachWithoutPDN-Connectivity? in RRCConnectionSetupComplete-NB
    //THEN the ESM DUMMY MESSAGE is piggybacked in ATTACH REQUEST, OTHERWISE the PDN CONNECTIVITY REQUEST message is piggybacked in ATTACH REQUEST.
    v_NasInd := f_NBIOT_RRC_ConnEst_DefWithNas(nbiot_Cell55,
                                               cr_EstablishmentCause_NB_Any,
                                               cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected, //WA#22_5_14 tsc_SHT_NoSecurityProtection,
                                                                                 //WA#22_5_14 cdr_CIOT_ATTACH_REQUEST_WithIMSINoTai(cr_AdditionalUpdateType_CIOT),
                                                                                 cr_CIOT_ATTACH_REQUEST(cr_AdditionalUpdateType_CIOT),
                                                                                 v_ESMMessage));
    //@siclog "Step 65" siclog@
    //The SS transmits an ATTACH REJECT message, EMM cause = " No suitable cells in tracking area ".
    //(The list of "forbidden tracking areas for roaming" in the UE should now contain TAI-7)
    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell55,
                                   tsc_SRB1bis,
                                   cs_TimingInfo_Now,
                                   cs_NAS_Request(f_GetEPS_SecurityHeaderForReject(tsc_Cause_NoSuitableCells), // @sic R5-174501 sic@
                                                  cs_508_AttachReject(cs_EMM_Cause_v(tsc_Cause_NoSuitableCells)))));
    //@siclog "Step 65A" siclog@
…
…
    //@siclog "Step 69" siclog@
    //Check: Does the UE transmit an ATTACH REQUEST message in the next 30 seconds on Ncell 55 or Ncell 62?    
    f_NBIOT_CheckNoAttachOn2Cells(nbiot_Cell55, nbiot_Cell62, 30.0, "Step 69"); //WA#22_5_14 changed "Step 67" -> "Step 69"
    //@siclog "Step 70" siclog@
    //If possible (see ICS) switch off is performed or the USIM is removed.
    //Otherwise the power is removed.
    f_UT_SwitchOffUE(UT, true);
    //@siclog "Step 71" siclog@
    //SS adjusts the cell power levels according to row T12 in table 22.5.14.3-1
    //NCell50 and NCell51 will not be config power level since it is not used anymore
    v_CellPowerList := {
      cs_NB_CellPower(nbiot_Cell55, tsc_ServingCellRS_EPRE),
      cs_NB_CellPower(nbiot_Cell56, tsc_NonSuitableOffNBIOTCellRS_EPRE),
      cs_NB_CellPower(nbiot_Cell62, tsc_NonSuitableOffNBIOTCellRS_EPRE)
    };
    f_NBIOT_SetCellPowerList(v_CellPowerList);
    f_NBIOT_ReleaseCellSpecific(nbiot_Cell62);//NBIOT Cell to be released as soon as it is not used anymore
    //@siclog "Step 72" siclog@
    //The UE is brought back to operation or the USIM is inserted.
    f_UT_SwitchOnUE(UT, true);    
    //@siclog "Step 73" siclog@
    //The UE transmits an ATTACH REQUEST message on Ncell 55.
    //IF AttachWithoutPDN AND UE sets attachWithoutPDN-Connectivity? in RRCConnectionSetupComplete-NB
    //THEN the ESM DUMMY MESSAGE is piggybacked in ATTACH REQUEST, OTHERWISE the PDN CONNECTIVITY REQUEST message is piggybacked in ATTACH REQUEST.
    v_NasInd := f_NBIOT_RRC_ConnEst_DefWithNas(nbiot_Cell55,
                                               cr_EstablishmentCause_NB_Any,
                                               cr_NAS_IndicationWithPiggybacking(tsc_SHT_NoSecurityProtection,
                                                                                 cdr_CIOT_ATTACH_REQUEST_WithIMSINoTai(cr_AdditionalUpdateType_CIOT),
                                                                                 v_ESMMessage));
    //@siclog "Step 74" siclog@
    //SS adjusts the cell power levels according to row T13 in table 22.5.14.3-1
    /*WA#22_5_14 Prose CR required
    //NCell50 and NCell51 and NCell62 will not be config power level since it is not used anymore
    v_CellPowerList := {
      cs_NB_CellPower(nbiot_Cell55, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE),
      cs_NB_CellPower(nbiot_Cell56, tsc_ServingCellRS_EPRE)
    };
    f_NBIOT_SetCellPowerList(v_CellPowerList);
    */
    //@siclog "Step 74A - 74D" siclog@
    //Steps 5 to 8 from procedure 8.1.5.2.3 in TS 36.508.
    v_NasCountUL := f_NBIOT_InitialRegistration_Step5_8 (nbiot_Cell55, //WA#22_5_14 nbiot_Cell56
                                                         v_NasInd,
                                                         false);//@sic R5-176875 sic@
    //@siclog "Step 75" siclog@
    //The SS transmits an ATTACH REJECT message,
    //EMM cause = No suitable cells in tracking area?.
    //(The list of forbidden tracking areas for roaming? in the UE should now contain TAI-9)
    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell55, //WA#22_5_14 nbiot_Cell56
                                   tsc_SRB1bis,
                                   cs_TimingInfo_Now,
                                   cs_NAS_Request(f_GetEPS_SecurityHeaderForReject(tsc_Cause_NoSuitableCells), // @sic R5-174501 sic@
                                                  cs_508_AttachReject(cs_EMM_Cause_v(tsc_Cause_NoSuitableCells)))));
    //@siclog "Step 76" siclog@
    //The SS releases the RRC connection.
    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell55);
    //WA#22_5_14
    //NCell50 and NCell51 and NCell62 will not be config power level since it is not used anymore
    v_CellPowerList := {
      cs_NB_CellPower(nbiot_Cell55, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE),
      cs_NB_CellPower(nbiot_Cell56, tsc_ServingCellRS_EPRE)
    };
    f_NBIOT_SetCellPowerList(v_CellPowerList);
    //@siclog "Step 77" siclog@
    //Check: Does the UE transmit an ATTACH REQUEST message on Ncell 56?
    //IF AttachWithoutPDN AND UE sets attachWithoutPDN-Connectivity? in RRCConnectionSetupComplete-NB
    //THEN the ESM DUMMY MESSAGE is piggybacked in ATTACH REQUEST, OTHERWISE the PDN CONNECTIVITY REQUEST message is piggybacked in ATTACH REQUEST.    
    v_NasInd := f_NBIOT_RRC_ConnEst_DefWithNas(nbiot_Cell56,
                                               cr_EstablishmentCause_NB_Any,
                                               cr_NAS_IndicationWithPiggybacking(tsc_SHT_NoSecurityProtection,
                                                                                 cdr_CIOT_ATTACH_REQUEST_WithIMSINoTai(cr_AdditionalUpdateType_CIOT),
                                                                                 v_ESMMessage));
    //@siclog "Step 77A - 77D" siclog@
    //Steps 5 to 8 from procedure 8.1.5.2.3 in TS 36.508.
    v_NasCountUL := f_NBIOT_InitialRegistration_Step5_8 (nbiot_Cell56,
                                                         v_NasInd,
                                                         false);//@sic R5-176875 sic@
    //@siclog "Step 78" siclog@
    //The SS transmits an ATTACH REJECT message, EMM cause = No suitable cells in tracking area?.
    //(The list of forbidden tracking areas for roaming? in the UE should now contain TAI-7 and TAI-8)
    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell56,
                                   tsc_SRB1bis,
                                   cs_TimingInfo_Now,
                                   cs_NAS_Request(f_GetEPS_SecurityHeaderForReject(tsc_Cause_NoSuitableCells), // @sic R5-174501 sic@
                                                  cs_508_AttachReject(cs_EMM_Cause_v(tsc_Cause_NoSuitableCells)))));
    //@siclog "Step 79" siclog@
    //The SS releases the RRC connection.
    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell56);
    //@siclog "Step 80" siclog@
    //SS adjusts the cell power levels according to row T14 in table 22.5.14.3-1
    //NCell50 and NCell51 and NCell62 will not be config power level since it is not used anymore
    v_CellPowerList := {
      //WA#22_5_14 cs_NB_CellPower(nbiot_Cell50, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE),//@sic R5-176875 sic@
      cs_NB_CellPower(nbiot_Cell55, tsc_ServingCellRS_EPRE),
      cs_NB_CellPower(nbiot_Cell56, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE)
    };
    f_NBIOT_SetCellPowerList(v_CellPowerList);
    //The following messages are to be observed on Ncell 55 unless explicitly stated otherwise.
    //@siclog "Step 81" siclog@
    //Check: Does the UE transmit the ATTACH REQUEST message in the next 30 seconds?    
    f_NBIOT_CheckNoAttachOn1Cell(nbiot_Cell55, 30.0, "Step 81");
    //@siclog "Step 82" siclog@
    //If possible (see ICS) switch off is performed. Otherwise the power is removed.
    f_UT_SwitchOffUE(UT, true);
    //@siclog "Step 83" siclog@
    //The UE is brought back to operation.
    f_UT_SwitchOnUE(UT, true);
    //@siclog "Step 84" siclog@
    //Check: Does the UE transmit an ATTACH REQUEST?
    //IF AttachWithoutPDN AND UE sets attachWithoutPDN-Connectivity? in RRCConnectionSetupComplete-NB
    //THEN the ESM DUMMY MESSAGE is piggybacked in ATTACH REQUEST, OTHERWISE the PDN CONNECTIVITY REQUEST message is piggybacked in ATTACH REQUEST.
    v_NasInd := f_NBIOT_RRC_ConnEst_DefWithNas(nbiot_Cell55,
                                               cr_EstablishmentCause_NB_Any,
                                               cr_NAS_IndicationWithPiggybacking(tsc_SHT_NoSecurityProtection,
                                                                                 cdr_CIOT_ATTACH_REQUEST_WithIMSINoTai(cr_AdditionalUpdateType_CIOT),
                                                                                 v_ESMMessage));
    //@siclog "Step 85 - 95" siclog@
    //The attach procedure is completed by executing steps 5 to 14b1 of the UE registration procedure in TS 36.508 sub clause 8.1.5.2.
    f_NBIOT_InitialRegistration_Step5_14(nbiot_Cell55, CONTROL_PLANE, STATE2_NB_CONNECTED_UPDATED, v_NasInd);
    //At the end of this test procedure sequence, the UE is in end state 2-NB Connected Mode according to TS 36.508.
  }


4.3 Change – function f_TC_22_5_14_NBIOT()
	Function name
	f_TC_22_5_14_NBIOT()

	Reason for change
	As per the Prose 36.523-1 Sec 22.5.14.3.1 Pre-test conditions for system simulator, the testcase should be using combination c2. However, the current TTCN uses c3. This needs to be corrected.
Also, the testcase in later stages uses Ncell 63, but this is not created at all. This also needs to be implemented.

	Summary of change
	Instead of calling function f_NBIOT_Init(c3, CONTROL_PLANE, LTE_CellsOnDifferentPLMN);, called f_NBIOT_Init(c2, CONTROL_PLANE, LTE_CellsOnDifferentPLMN);
Called function f_NBIOT_CellConfig_Def(nbiot_Cell63, CONTROL_PLANE);. Please see below

	TTCN module
	NBIOT_NAS_Attach.ttcn

	MCC160 Comment
	


Before change
	function f_TC_22_5_14_NBIOT() runs on NBIOT_PTC
  { /* NB-IoT / Attach / Rejected / Tracking Area not allowed/Roaming not allowed in this tracking area / No suitable cells in tracking area */
    f_NBIOT_Init(c3, CONTROL_PLANE, LTE_CellsOnDifferentPLMN);
    // create cell and send out system information
    f_NBIOT_CellConfig_Def(nbiot_Cell50, CONTROL_PLANE);//PLMN1 HPLMN
    f_NBIOT_CellConfig_Def(nbiot_Cell51, CONTROL_PLANE);
    f_NBIOT_CellConfig_Def(nbiot_Cell55, CONTROL_PLANE);
    f_NBIOT_CellConfig_Def(nbiot_Cell56, CONTROL_PLANE);
    f_NBIOT_CellConfig_Def(nbiot_Cell62, CONTROL_PLANE);
    // Preamble: The UE is in State 1-NB "Switched OFF"
    // ==========================================================================
    f_NBIOT_Preamble(nbiot_Cell50, CONTROL_PLANE);
    f_NBIOT_UE_Detach_SwitchOffUe(nbiot_Cell50, CONTROL_PLANE, RRC_CONNECTED);
    f_NBIOT_TestBody_Set(true);
    fl_TC_22_5_14_Body();
    // Postamble
    f_NBIOT_TestBody_Set(false);
    f_NBIOT_Postamble (nbiot_Cell55, CONTROL_PLANE, N2_CONNECTED);
  }


After change
	function f_TC_22_5_14_NBIOT() runs on NBIOT_PTC
  { /* NB-IoT / Attach / Rejected / Tracking Area not allowed/Roaming not allowed in this tracking area / No suitable cells in tracking area */
    f_NBIOT_Init(c2, CONTROL_PLANE, LTE_CellsOnDifferentPLMN); //WA#22_5_14
    // create cell and send out system information
    f_NBIOT_CellConfig_Def(nbiot_Cell50, CONTROL_PLANE);//PLMN1 HPLMN
    f_NBIOT_CellConfig_Def(nbiot_Cell51, CONTROL_PLANE);
    f_NBIOT_CellConfig_Def(nbiot_Cell55, CONTROL_PLANE);
    f_NBIOT_CellConfig_Def(nbiot_Cell56, CONTROL_PLANE);
    f_NBIOT_CellConfig_Def(nbiot_Cell62, CONTROL_PLANE);
    f_NBIOT_CellConfig_Def(nbiot_Cell63, CONTROL_PLANE); //WA#22_5_14
    // Preamble: The UE is in State 1-NB "Switched OFF"
    // ==========================================================================
    f_NBIOT_Preamble(nbiot_Cell50, CONTROL_PLANE);
    f_NBIOT_UE_Detach_SwitchOffUe(nbiot_Cell50, CONTROL_PLANE, RRC_CONNECTED);
    f_NBIOT_TestBody_Set(true);
    fl_TC_22_5_14_Body();
    // Postamble
    f_NBIOT_TestBody_Set(false);
    f_NBIOT_Postamble (nbiot_Cell55, CONTROL_PLANE, N2_CONNECTED);
  }


4.4 Change – function TC_22_5_14()

	Function name
	TC_22_5_14()

	Reason for change
	The guard timer for this testcase is 300s, which is too short. This should be extended to 3600s.

	Summary of change
	Changed the guard timer to 3600s instead of 300s.

	TTCN module
	NBIOT_Testsuite.ttcn

	MCC160 Comment
	


Before change
	testcase TC_22_5_14() runs on MTC_NBIOT system SYSTEM_NBIOT {
    // @purpose
    //   NB-IoT / Attach / Rejected / Tracking Area not allowed / Roaming not allowed in this tracking area / No suitable cells in tracking area
    var NBIOT_PTC        v_NBIOT      := null;
    timer t_GuardTimer := int2float(300);

    v_NBIOT := NBIOT_PTC.create alive;
    f_MTC_ConnectPTCs_NBIOT(system, v_NBIOT);
    v_NBIOT.start(f_TC_22_5_14_NBIOT());
    t_GuardTimer.start;
    f_MTC_MainLoop_NBIOT(t_GuardTimer);
  }


After change
	testcase TC_22_5_14() runs on MTC_NBIOT system SYSTEM_NBIOT {
    // @purpose
    //   NB-IoT / Attach / Rejected / Tracking Area not allowed / Roaming not allowed in this tracking area / No suitable cells in tracking area
    var NBIOT_PTC        v_NBIOT      := null;
    timer t_GuardTimer := int2float(3600); //WA#22_5_14 300->3600
    v_NBIOT := NBIOT_PTC.create alive;
    f_MTC_ConnectPTCs_NBIOT(system, v_NBIOT);
    v_NBIOT.start(f_TC_22_5_14_NBIOT());
    t_GuardTimer.start;
    f_MTC_MainLoop_NBIOT(t_GuardTimer);
  }


5 Branches executed in test case 22.5.14
The test case implementation was executed with Integrity and ciphering enabled.

6 Execution Log Files

6.1 Hisilicon Hi2110
The Hisilicon device passed this test case on Rohde & Schwarz CMW500. The documentation below is enclosed as evidence of the successful test case run [1]:

Test Case Execution log file: 
Rohde-Schwarz\tc_22_5_14_Hisilicon.log 
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)
In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file
6.2 MTK_2625

The MTK UE passed this NB-IoT in LTE FDD band 3 on Anritsu Protocol Conformance Test System ME7834LA.The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file

(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)


In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

